

情報科学演習  
第 12 回  
数式処理系 Maple 入門

目 次

|       |                                 |   |
|-------|---------------------------------|---|
| 1     | はじめに (Maple とは)                 | 1 |
| 2     | Maple の起動と簡単な計算                 | 1 |
| 2.1   | 簡単な計算 . . . . .                 | 1 |
| 2.1.1 | 様々な定数, 手続き (関数) . . . . .       | 2 |
| 2.1.2 | パッケージ (package) を利用する . . . . . | 4 |
| 2.2   | グラフィックス機能 . . . . .             | 4 |
| 2.3   | Help . . . . .                  | 5 |
| 3     | 補足                              | 5 |

## 1 はじめに (Maple とは)

今回は数式処理系 Maple を紹介します。数式処理とは、数学で行う計算、即ち、式の展開、因数分解、微分、積分、方程式を解く、逆行列を求める、固有値計算等を数式のままコンピュータでする事を指します。実際 Maple, Mathematica, Macsyma (MAXIMA) 等の数式処理ソフトは、大学初年級以上の数学を処理する能力を持っており、数学教育で用いるソフトとして便利です、研究上の実験ソフトとしても役に立ちます。

Maple は、Canada の Waterloo 大学で開発された数式処理ソフトです。Maple は、沖縄では見ませんが、楓という木の事です。Canada 国旗の中央部には楓の葉がデザインされています<sup>1</sup>。現在、Waterloo Maple Inc. が販売しております。サポートされている OS は、MacOS, Windows, Solaris, Linux (i386) 等です。

Maple の開発は、1980 年頃に始まったようです。私が初めて Maple に触れたのが 1987 年位で、当時の Version が 4.7 でした。この頃、ようやくパーソナルコンピュータ (Mac) で、Maple が動くようになりました。

このテキストでの約束:

このテキストでは、> で始まっている行は、Maple の入力です。改行部分ではエンターキーを押します。

総合情報処理センターのほとんどのマシンに、Maple が入っています。Maple は一般ユーザ版は 20 万円以上しますが、大学生協を通して学生版を買うと、21,000 円で購入出来ます。詳しくは、次を参照して下さい。

[http://www.cybernet.co.jp/maple/contact/price\\_student.shtml](http://www.cybernet.co.jp/maple/contact/price_student.shtml)

Maple は、linux でも動くのですが、情報処理センターには linux 版が入っていません。今回は Windows XP を起動して logon して下さい。

## 2 Maple の起動と簡単な計算

Windows のデスクトップに「Applications」と言うフォルダがあるはずです。それをダブルクリックすると中に「Maple 11 ショートカット」と言うアイコンがありますから、それをダブルクリックして下さい。メニューはそれを読めば、機能は大体理解できます。ここではそれを使わないので、説明は省略します。

### 2.1 簡単な計算

ウィンドウ内に赤く不等号 > がでている部分があります。ここに数式を入力する事で計算ができます。メニューバーの File メニューの New を使えば、このウィンドウは新たに作る事もできます。

Maple の四則演算の記号はそれぞれ、+, -, \*, / です。Maple では冪乗と階乗が定義されており、それぞれ、^ と ! を用います。Maple では、文の最後をセミコロン ; で終ってエンターキーを押しますと、文の評価結果が出力されます。エンターキーだけでは単なる改行となります。文中の改行は無視されます。文中の括弧 ( ) は数学と同じ意味になります。四則演算の優先順位も数学と一致します。次を実行してみてください。

```
> 1 + 2;  
> 10/3 + 2;  
> 10/3.0;  
> 2^10;  
> 50!;  
> (a+a-b)*c/d;  
> a^2 + a;
```

有理数の扱い、文字式の扱いが数学と一致します。小数が式に含まれていれば、自動的に小数扱いされます。

<sup>1</sup> 個人的には、ホットケイキのシロップは、メイプルシロップ。

### 2.1.1 様々な定数, 手続き (函数)

Maple では, 数式処理のための手続きが 2700 以上定義されています. それらを全部解説する事は不可能ですので, ここではその一例をあげます. これらの例の中に初等函数が用いられていますが, それらの意味は容易に類推できると思いますので, これについての解説はいちいちしません.

数式処理に欠かせない定数 (円周率等) が既に定義されています. 円周率は, Pi という記号を使います.

```
> Pi;  
> cos(Pi/4);  
> tan(Pi/2);  
> arctan(-infinity);
```

$\sqrt{2}$  がそのまま出て来る事, tan の不定値に対するエラーメッセージに注意して下さい. arctan は逆正接函数, infinity は無限大の事です. 最後の答は, 極限值を出力しています.

Maple では, 有理数, 冪根, 円周率等の定数は, そのまま出力されます. 上の逆正接函数の計算でもそうですし, 例えば,  $1 + \frac{1}{2^2} + \frac{1}{3^2} + \cdots = \sum_{n=1}^{\infty} \frac{1}{n^2} = \zeta(2) = \frac{\pi^2}{6}$  も次で計算させると, 円周率を使った答が出ます.

```
> sum(1/n^2, n=1..infinity);
```

これを小数へ変換するには, evalf という手続きを用います.

```
> evalf(Pi^2/6);
```

Maple では, 非常に正確な数値計算ができます. 例えば,  $e^{\pi\sqrt{163}} - 744 - 640320^3$  を C 言語付属の数学関数ライブラリで計算しますと -480 という答を得ますが, Maple を使うと, C 言語の計算がとんでもない誤差を含んでいる事がわかります. 起動時のデフォルトでは, 浮動小数点の仮数部は 10 桁に設定されており, そのままで計算しますと, この結果は真の値の約 40 倍という, やはりとんでもない答が返って来ますので, 仮数部の桁数を事前に設定します. 仮数部の桁数は, Maple のシステム変数 Digits に格納されていますから, この値を変更します. Maple では代入には := を用います.

```
> Digits:=50;  
> evalf(exp(Pi*sqrt(163))-744-640320^3);
```

C 言語のライブラリを用いた計算が, 真の値の  $6 \times 10^{14}$  倍以上の値になっている事がわかります.

課題 (難):  $e^{\sqrt{163}\pi}$  の値が整数に近い理由を調べよ. 菅にこれを説明できる人は単位を A であげます. (同じ理由で,  $e^{\sqrt{67}\pi}$  も整数に近い.)

文字式の展開, 因数分解も可能です.

```
> expand((x+y)^5);  
> factor(a^8-b^8);
```

次の問題は, 2000 年の琉球大学入学試験問題前期日程数学甲の 1 です.

1. 関数  $\frac{x}{\sqrt{1+x^2}}$  を微分せよ.

2. 不定積分  $\int x\sqrt{x^2+2} \, dx$  を求めよ.

3. 定積分  $\int_{\frac{1}{e}}^{2e} x^3 \log x \, dx$  を求めよ.

4.  $\lim_{x \rightarrow 0} \frac{\sqrt{2x+1} - 1 - x}{x^2}$  を求めよ.

このような単純な計算は, Maple は得意です. (試験では, 答だけを書いても満点にはならないと思います.) Maple V では, 直前の結果を % で参照できます.

```
> diff(x/sqrt(1+x^2), x);
> simplify(%);
> int(x*sqrt(x^2+2), x);
> int(x^3*log(x), x=exp(-1)..2*exp(1));
> limit((sqrt(2*x+1)-1-x)/x^2, x=0);
```

方程式  $f(x) = 0$  の解を求める様々な方法も Maple には用意されています.  $f(x)$  が 4 次以下の多項式なら, この方程式には代数的な解法が存在する事が知られています. (3 年の代数学 I・II で勉強する予定です.) Maple はこれらの解法を知っており, solve という手続きになっています. 次を実行してみてください. これらの解には複素数が含まれますが, Maple では虚数単位は大文字の I で表示されます.

```
> solve(x^3+1, x);
> solve(x^3+3*x+1, x);
```

5 次以上の方程式には, 代数的な解法が一般には存在しない事が知られています. (代数的という制限を外せば, 別な解法はあります.) 次を実行してみてください.

```
> solve(x^5+x^2+1, x);
```

RootOf( $_Z^5 + _Z^2 + 1$ ) という解が出て来ます. もちろん, これは単なるトートロジーに過ぎないのですが, Maple は代数的数を扱えるので, この解 (代数的数) を用いた計算が今後記号的に可能です.

代数的な解法がある場合でも, その解法が複雑な場合には, 残念ながら代数的な解を出力しません. 例えば, 1 の 7 乗根を計算させようとしても, de Moivre の公式から出てくる解が単純に出力されるだけです.

```
> solve(x^7-1, x);
```

上で述べたように Maple では, 代数的数が扱えます. これを利用すると,  $x^7 - 1 = 0$  の代数的な解も求める事ができます.  $x^7 - 1 = (x - 1)(x^6 + x^5 + x^4 + x^3 + x^2 + x + 1)$  ですが, 積の右側の 6 次式は  $\sqrt{-7}$  を使うと, 2 つの 3 次式の積に因数分解されます. この様な因数分解は, 付け加える数を factor の第 2 引数に加える事で可能です.

```
> factor(x^6+x^5+x^4+x^3+x^2+x+1, (-7)^(1/2));
```

従って, 1 の複素 7 乗根は, 次の 2 つ方程式の解全体です.

```
> solve(2*x^3+x^2-I*sqrt(7)*x^2-x-I*sqrt(7)*x-2, x);
> solve(2*x^3+x^2+I*sqrt(7)*x^2-x+I*sqrt(7)*x-2, x);
```

$f(x)$  が 5 次以上の多項式や, 多項式以外の場合には, 一般的な解法が存在しません. このような方程式の近似解を数値的に計算する方法も Maple は知っています. 上の方程式の数値解も, 次で計算してくれます.

```
> fsolve(x^5+x^2+1, x);
```

実際, 関数のグラフを描画する plot を使って,

```
> plot(x^5+x^2+1,x=-5..5);
> plot(x^5+x^2+1,x=-2..0);
```

等を実行しますと, `fsolve` で求めた値が  $x^5 + x^2 + 1 = 0$  の唯 1 の実数解の近似値である事がわかります.

### 2.1.2 パッケージ (package) を利用する

これまでの計算では, 特別な事をせずともそれを実行しますが, 行列の計算などはそのままでは実行してくれず, それを計算するためのパッケージ (ライブラリ) を呼び出す必要があります. ここでは, 線形代数学のパッケージを呼び出して, そこで定義されている計算をしてみます.

パッケージを読み込むには `with` という手続きを使います. 線形代数学のパッケージ名は `linalg` となっておりますので, 次のように入力します.

```
> with(linalg);
```

この時に出力されるのが, `linalg` のパッケージで定義されている手続き名です. 一例をあげますと, 次のようになります.

```
> B:= matrix(2,2);
> trace(B);
> det(B);
> inverse(B);
> eigenvalues(B);
```

どのようなパッケージが利用できるかは, 手続き `help` に引数 `package` をいれて利用する事でわかります.

```
> help(package);
```

## 2.2 グラフィックス機能

上でも述べたように, 例えば,  $\sin x$  の関数のグラフを書くには, 次のように入力します.

```
> plot(sin(x), x = -Pi .. Pi);
```

2 変数関数のグラフも簡単に書けます.

```
> plot3d(cos(2*x^3+y^2), x=-2..2, y=-2..2);
```

マウスの左ボタンで図形を掴んで動かせば, 立方体が動きます. 動かした後, 右ボタンを押して `Redraw` を選べば, 動かした図形が再描画されます.

Maple では, 座標を順に与えてそれを線分で結び多角形を描く事ができます. まずは, グラフィックス表示のためのパッケージ `plots` を読み込みます.

```
> with(plots);
```

次を実行して見て下さい.

```
> cornercoordinates := [[0,0],[1,0],[1,-1],[0,-1]]:
> asquare := polygonplot(cornercoordinates):
> display({asquare}, view=[0..3, 0..-2], scaling=CONSTRAINED);
> ngon := n -> [seq([cos(2*Pi*i/n),sin(2*Pi*i/n)], i = 1..n)]:
> display([polygonplot(ngon(8))]);
> fivestar:= [seq([cos(2*Pi*(2*i+1)/5),sin(2*Pi*(2*i+1)/5)], i = 1..5)]:
> display([polygonplot](fivestar));
```

## 2.3 Help

Maple 本体のウィンドウのメニューバーの右端にヘルプメニューがあります。ここからさまざまな機能を知る事ができます。Maple は非常に多機能です。ヘルプブラウザーで必要な項目にたどり着くのも大変です。ブラウザー以外にもヘルプメニューの Topic Search, Full Text Search で検索することができます。さらに、上で述べたように Maple のプロンプト行で

> ? キーワード

あるいは

> help(キーワード);

とすると、Topic Search とほぼ同じ事が実行されます。次でも、利用上役立つ様々な情報が得られます。

> help(help);

ヘルプメニューの中に「Maple ツアー」と言うのがあります。Maple の使い方が一通り説明されます。残った時間はこの Maple ツアーをやってみて下さい。また、微積や線形代数の教科書の問題を Maple で解いてみて下さい。(結構間違いがあったりして...)

2002 年の計算機言語 I で Maple を用いたプログラミングの講義を行いました。そのときのテキストが (L<sup>A</sup>T<sub>E</sub>X ファイルですが)、次の場所にあります。使い方をより知りたい方は、参考にして下さい。Maple の version が当時から大分上がっていますが、内容はほぼ今でも通用するはずで、来年の計算機概論 I でも、もう一度 Maple を取り上げる予定です。今回のテキストは、そこにある 1.tex を改訂した物です。

<ftp://ftp.math.u-ryukyu.ac.jp/pub/gengo/2002/>

## 3 補足

Windows のデスクトップで、「ホームディレクトリ」と書かれた物が、左上角にありますが、これは Vine の「xxx のホーム」を見たときの「WIN」と言うフォルダです。情報処理センターのシステムでは、Windows では、この「ホームディレクトリ」の中だけに、個人のファイルを作る事が出来ます。

この講義で作っている、個人の Web ページを置くフォルダ「WWW」は、Windows からは直接見る事が出来ません。Windows で Web ページを作成した場合、ファイル転送ソフトを用いるか、リモート端末ソフトでサーバに login してファイルを移動するかをする必要があります。

次週は、もう一度 Web ページ作成に戻ります。以前に言っておいた、オリジナルページを作ってください。で取材した材料を持って来て下さい。成績評価の基準となる Web ページ完成締切は 8 月 10 日とします。

## 参考文献

- [1] B. W. Char 他, サイバネットシステム訳, はじめての Maple, 1998, シュプリンガーフェアラーク東京
- [2] B. W. Char 他, サイバネットシステム訳, よくわかる Maple, 1998, シュプリンガーフェアラーク東京
- [3] K. M. Heal, M. Hansen, K. Rickard 著, 示野信一他訳, Maple V Release 5 ラーニングガイド, 1999, シュプリンガーフェアラーク東京
- [4] 示野信一著, Maple V で見る数学ワールド, 1999, シュプリンガーフェアラーク東京